

CUESTIÓN 1 (5 puntos)

Se quiere diseñar un contador síncrono de dos bits con inicialización síncrona (**clear**) y asíncrona (**reset**), señal de habilitación (**enable**) y que ofrezca como salidas una señal de tipo *std_logic_vector* de dos bits con la cuenta (**salida**) y una señal de acarreo (**cout**).

Se piden los siguientes códigos VHDL:

- a) Entidad del contador (0.5 puntos).
- b) Arquitectura del contador, usando una estructura de contador binario y una señal de tipo *unsigned* como elemento de almacenamiento. (1 punto)
- c) Arquitectura del contador, de tipo estructural. En este caso se asumirá que existe una entidad y una arquitectura de un contador síncrono de un bit ya diseñada, con el que formar el contador de dos bits pedido. (1 punto)
- d) Arquitectura del contador usando un contador en anillo como circuito de cuenta, pero que tenga la misma salida que los casos b) y c). (1 punto)
- e) Arquitectura del contador diseñado mediante una codificación similar a una máquina de estado en la que se use una señal enumerada que describe los estados (E0, E1, E2, ...) como el valor de cuenta interno, y haciendo que la señal salida siga una cuenta igual a un código Gray (es decir, progresiva y cíclica). (1,5 punto)

CUESTIÓN 2 (5 puntos)

Se quiere diseñar un generador de señales digitales configurable basado en un registro de 32 bits que denominaremos acumulador de fase. La salida, que será una señal (**wave**) de 8 bits, se podrá elegir entre cuatro tipos distintos de formas de onda: dientes de sierra, triangular, cuadrada, y PWM. Las formas de onda cuadrada y PWM sólo podrán tener dos valores de salida: 0x00 y 0xff.

El circuito contará con las siguientes entradas de control:

- Incremento (**inc**, 8 bits), con la que se controla la frecuencia de la señal generada. Cada ciclo de reloj, el circuito a diseñar sumará este valor al registro acumulador de fase.
- Selector de forma de onda (**sel**, 2 bits). Queda a vuestra elección la forma de onda que se generará con cada combinación de bits en esta señal.
- Ciclo de trabajo (**duty**, 8 bits) de la señal PWM.

Las distintas formas de onda se generarán de acuerdo con los siguientes criterios:

- La forma de onda de **dientes de sierra** (rampa ascendente) se corresponderá con los 8 bits más significativos del acumulador de fase.
- La forma de onda **triangular** se generará haciendo una operación XOR, bit a bit, entre el bit más significativo del acumulador de fase y cada uno de los 8 bits siguientes (*).
- La forma de onda **cuadrada** se generará con el bit más significativo del acumulador de fase.
- La forma de onda **PWM** se generará comparando los 8 bits más significativos del acumulador de fase con la entrada de control **duty**, que fija el ciclo de trabajo de la señal.

(*) La subida corresponde a los valores en los que el bit más significativo vale '0', y por lo tanto '0' XOR bit(i) = bit(i). La bajada será en los valores en los que el bit más significativo vale '1', por lo que '1' XOR bit(i) = NOT bit(i).

Desarrollar el código VHDL del circuito descrito anteriormente, incluyendo entidad y arquitectura.

Duración del examen: 1 hora 30 minutos

```

1  library ieee;
2  use ieee.std_logic_1164.all;
3  use ieee.numeric_std.all;
4
5  entity ej1 is
6      port (
7          clk      : in  std_logic;
8          reset    : in  std_logic;
9          clear     : in  std_logic;
10         enable   : in  std_logic;
11         salida   : out std_logic_vector(1 downto 0);
12         cout     : out std_logic
13     );
14 end ej1;
15
16 architecture arch1 of ej1 is
17
18     signal cnt : unsigned(1 downto 0);
19
20 begin
21
22     process(clk,reset)
23     begin
24         if reset = '1' then
25             cnt <= (others => '0');
26         elsif clk'event and clk = '1' then
27             if clear = '1' then
28                 cnt <= (others => '0');
29             else
30                 if enable = '1' then
31                     if cnt = "11" then
32                         cnt <= (others => '0');
33                     else
34                         cnt <= cnt + 1;
35                     end if;
36                 end if;
37             end if;
38         end if;
39     end process;
40
41     salida <= std_logic_vector(cnt);
42     cout <= '1' when cnt = "11" and enable = '1' else '0';
43
44 end arch1;
45
46 architecture arch2 of ej1 is
47
48     component cnt is
49         port (
50             clk      : in  std_logic;
51             reset    : in  std_logic;
52             clear     : in  std_logic;
53             enable   : in  std_logic;
54             salida   : out std_logic;
55             cout     : out std_logic
56         );
57     end component;
58
59     signal carry : std_logic;
60
61 begin
62
63     bit0: cnt
64     port map (
65         clk      => clk,
66         reset    => reset,
67         clear     => clear,
68         enable   => enable,
69         salida   => salida(0),
70         cout     => carry
71     );
72
73     bit1: cnt

```

```

74     port map (
75         clk      => clk,
76         reset    => reset,
77         clear    => clear,
78         enable   => carry,
79         salida   => salida(1),
80         cout     => cout
81     );
82
83 end arch2;
84
85 architecture arch3 of ej1 is
86
87     signal ring : unsigned(3 downto 0);
88
89 begin
90
91     process(clk,reset)
92     begin
93         if reset = '1' then
94             ring <= (0 => '1', others => '0');
95         elsif clk'event and clk = '1' then
96             if clear = '1' then
97                 ring <= (0 => '1', others => '0');
98             else
99                 if enable = '1' then
100                     ring <= ring(2 downto 0) & ring(3);
101                 end if;
102             end if;
103         end if;
104     end process;
105
106     with ring select
107         salida <= "00" when "0001",
108                  "01" when "0010",
109                  "10" when "0100",
110                  "11" when "1000",
111                  "--" when others;
112
113     cout <= '1' when ring = "1000" and enable = '1' else '0';
114
115 end arch3;
116
117 architecture arch4 of ej1 is
118
119     type state_t is (E0, E1, E2, E3);
120     signal state : state_t;
121
122 begin
123
124     process(clk,reset)
125     begin
126         if reset = '1' then
127             state <= E0;
128         elsif clk'event and clk = '1' then
129             if clear = '1' then
130                 state <= E0;
131             else
132                 if enable = '1' then
133                     case state is
134                         when E0 => state <= E1;
135                         when E1 => state <= E2;
136                         when E2 => state <= E3;
137                         when E3 => state <= E0;
138                     end case;
139                 end if;
140             end if;
141         end if;
142     end process;
143
144     with state select
145         salida <= "00" when E0,
146                  "01" when E1,

```

```
147         "11" when E2,  
148         "10" when E3,  
149         "--" when others;  
150  
151     cout <= '1' when state = E3 and enable = '1' else '0';  
152  
153 end arch4;  
154
```

```

1  library ieee;
2  use ieee.std_logic_1164.all;
3  use ieee.numeric_std.all;
4
5  entity ej2 is
6      port (
7          clk      : in  std_logic;
8          reset    : in  std_logic;
9          sel      : in  std_logic_vector(1 downto 0);
10         inc      : in  std_logic_vector(7 downto 0);
11         duty     : in  std_logic_vector(7 downto 0);
12         wave     : out std_logic_vector(7 downto 0)
13     );
14 end ej2;
15
16 architecture behavioral of ej2 is
17
18     -- Phase accumulator
19     signal acc : unsigned(31 downto 0);
20     -- Waveform: sawtooth
21     signal saw : std_logic_vector(7 downto 0);
22     -- Waveform: triangle
23     signal msbN : std_logic_vector(7 downto 0);
24     signal tri  : std_logic_vector(7 downto 0);
25     -- Waveform: square
26     signal sqr  : std_logic_vector(7 downto 0);
27     -- Waveform: PWM
28     signal pwm  : std_logic_vector(7 downto 0);
29
30 begin
31
32     -- Phase accumulator
33     process(clk, reset)
34     begin
35         if reset = '1' then
36             acc <= (others => '0');
37         elsif clk'event and clk = '1' then
38             acc <= acc + unsigned(inc);
39         end if;
40     end process;
41
42     -- Waveform: sawtooth
43     saw <= std_logic_vector(acc(31 downto 24));
44
45     -- Waveform: triangle
46     msbN <= (others => acc(31));
47     tri <= std_logic_vector(acc(30 downto 23)) xor msbN;
48
49     -- Waveform: square
50     sqr <= (others => not acc(31));
51
52     -- Waveform: PWM
53     pwm <= (others => '1') when acc(31 downto 24) < unsigned(duty) else (others =>
54         '0');
55
56     -- Output MUX
57     with sel select
58     wave <= saw when "00",
59           tri when "01",
60           sqr when "10",
61           pwm when "11",
62           (others => '-') when others;
63 end behavioral;
64

```